Geometry of phase space and solutions of semilinear elliptic equations in a ball

Jean Dolbeault and Isabel Flores

Paramètres généraux

```
n = 5;
p = (n + 2)/(n - 2) + 0.3;
α = n - 2 - 4/(p - 1);
β = 2/(p - 1) (n - 2 - 2/(p - 1));
Off[ParametricPlot3D::"ppcom"]
Off[ParametricPlot::"ppcom"]
```

Figure 1

```
kp = (-α + Sqrt[α^2 + 4 β])/2;
km = (-α - Sqrt[α^2 + 4 β])/2;
F1[x0_, y0_, t2_] :=
  ParametricPlot[Evaluate[{x[t], y[t]} /. NDSolve[{x'[t] == y[t], x[0] == x0,
       y'[t] == -α y[t] + β x[t] - Abs[x[t]]^(p-1) x[t], y[0] == y0}, {x, y}, {t, 0, t2}]],
   {t, 0, t2}, PlotRange → All, Ticks → None, DisplayFunction -> Identity];
F1a[ϵ_, t2_] := F1[ϵ, kp ϵ, t2/ϵ];
P1 = F1a[0.0001, 0.01];
P2 = F1[0, 2, 10];
P3 = F1[0, 3.66829365872, 25];
L[k_, xm_, xM_] := ParametricPlot[{x, k * x}, {x, -xm, xM},
   PlotRange → All, Ticks → None, DisplayFunction -> Identity];
L1 = L[kp, 0.2, 1];
L2 = L[km, 0.2, 1];
Show[{L1, L2, P1, P2, P3}, DisplayFunction -> $DisplayFunction];
```



Figure 2

```
F2[x0_, y0_, z0_, T_, c_] :=
 Evaluate[{x[T], y[T], c / 2} /. NDSolve[{x′[t] == y[t], x[0] == x0,
      y′[t] == -α y[t] + β x[t] - Abs[x[t]]^(p-1) x[t] - z[t] x[t], y[0] == y0,
      z′[t] == 2 z[t], z[0] == z0}, {x, y, z}, {t, 0, T}]]〚1〛
F2a[ϵ_, λ_, τ_] := Module[{P = F2[ϵ τ, kp ϵ τ, ϵ, 1/2 Log[λ/ϵ], λ]}, P]
F4[ϵ_, T_] :=
 Module[{P = Evaluate[{x[T], y[T], z[T]} /. NDSolve[{x'[t] == y[t], x[0] == β^(1/(p-1)),
        y'[t] == -α y[t] + β x[t] - Abs[x[t]]^(p-1) x[t] - z[t] x[t], y[0] == 0,
        z'[t] == 2 z[t], z[0] == ϵ}, {x, y, z}, {t, 0, T}]][[1]]}, P]
```

```
Q = ParametricPlot3D[F2a[0.0001, λ, τ], {λ, 0.1, 5}, {τ, 0, 500}, PlotRange →
    {{0, Automatic}, Automatic, Automatic}, Ticks → None, PlotPoints → {30, 50},
   LightSources → {{{1, 0, 0.5}, GrayLevel[1]}}, DisplayFunction → Identity];
UnsP = ParametricPlot3D[F4[0.01, τ], {τ, 0.1, 2.93}, Ticks → None,
  DisplayFunction → Identity]; LineP[A_, B_] :=
 ParametricPlot3D[t A + (1 - t) B, {t, 0, 1}, Ticks → None, DisplayFunction → Identity]
L1 = LineP[{0, 0, 0}, {2.3, 0, 0}];
L3 = LineP[{0, 0, 0}, {0, 0, 3}];
Show[Q, UnsP, L1, L3 , ViewPoint → {0.8, -0.7, 0.9}, DisplayFunction → $DisplayFunction];
```



Figure 3

```
F3[x0_, y0_, z0_, T_] := Evaluate[{x[T], y[T]} /. NDSolve[
     {x'[t] == y[t], x[0] == x0, y'[t] == -α y[t] + β x[t] - Abs[x[t]]^(p-1) x[t] - z[t] x[t],
      y[0] == y0, z'[t] == 2 z[t], z[0] == z0}, {x, y, z}, {t, 0, T}]][[1]]
F3a[ϵ_, λ_, τ_] := Module[{P = F3[ϵ τ, kp ϵ τ, ϵ, 1/2 Log[λ/ϵ]]}, P]
F3b[ϵ_, λ_] := Module[{P = F4[ϵ, 1/2 Log[λ/ϵ]]}, {P[[1]], P[[2]]}]
```

```
Show[P1, DisplayFunction -> $DisplayFunction];
```



```
Show[ParametricPlot[F3a[0.00000001, 5, τ], {τ, 0, 1000000},
   PlotRange → All, Ticks → None, PlotPoints → 300, DisplayFunction → Identity],
  Graphics[{PointSize[0.02], Point[F3b[0.01, 5]]}],
  DisplayFunction -> $DisplayFunction];
```



```
zstar = 0.01 * Exp[2 τ] /. FindRoot[F4[0.01, τ][[1]], {τ, 3, 3.5}]
Show[ParametricPlot[F3a[0.00000001, zstar, τ], {τ, 0, 1000000},
   PlotRange → All, Ticks → None, PlotPoints → 300, DisplayFunction → Identity],
  Graphics[{PointSize[0.02], Point[F3b[0.01, zstar]]}],
  DisplayFunction -> $DisplayFunction];
```

```
12.0675
```



```
Show[ParametricPlot[F3a[0.00000001, 20, τ], {τ, 0, 1000000},
   PlotRange → All, Ticks → None, PlotPoints → 300, DisplayFunction → Identity],
  Graphics[{PointSize[0.02], Point[F3b[0.01, 20]]}],
  DisplayFunction -> $DisplayFunction];
```



Figure 4

```
F4a[ϵ_, λ_, τ_] := Module[{P = F3[ϵ τ, kp ϵ τ, ϵ, 1/2 Log[λ/ϵ]]}, {P[[1]], P[[2]], Log[1 + λ]}]

UnsP = ParametricPlot3D[{F4[0.01, τ][[1]], F4[0.01, τ][[2]], Log[1 + F4[0.01, τ][[3]]]},
   {τ, 0.01, 4}, Ticks → None, DisplayFunction → Identity];
A1 = ParametricPlot3D[F4a[0.00000001, 0.1, τ], {τ, 0, 1000000}, PlotRange → All,
   Ticks → None, PlotPoints → 300, DisplayFunction → Identity];
A2 = ParametricPlot3D[F4a[0.00000001, 4, τ], {τ, 0, 10000}, PlotRange → All,
   Ticks → None, PlotPoints → 300, DisplayFunction → Identity];
A3 = ParametricPlot3D[F4a[0.00000001, 8, τ], {τ, 0, 10000}, PlotRange → All,
   Ticks → None, PlotPoints → 300, DisplayFunction → Identity];
LV = LineP[{0, 0, 0}, {0, 0, 3}];
LO = LineP[{0, 0, 0}, {2.3, 0, 0}];
LH1 = LineP[{0, -2, 0}, {0, 2, 0}];
LH2 = LineP[{0, -1.5, Log[1 + 4]}, {0, 1, Log[1 + 4]}];
LH3 = LineP[{0, -1, Log[1 + 8]}, {0, 1, Log[1 + 8]}];
Show[LV, LO, LH1, LH2, LH3, UnsP, A1, A2, A3,
  DisplayFunction -> $DisplayFunction, ViewPoint → {0.5, -1.1, 0.2}];
```



Figure 5

```
FL[a_, rmax_, eta_, d_, e_] :=
 Module[{G = FindRoot[u[x] == 0, {x, 2 * eta, 3.2}] /. NDSolve[{-v'[r] - (d - 1) v[r] / r ==
        u[r] + (Abs[u[r]]) ^ (e - 1) u[r], u'[r] == v[r], u[eta] == a - eta^2 (a + a^e) / (2 d),
       v[eta] == -eta (a + a^e) / d}, {u, v}, {r, eta, rmax}]},
  {x^2, Log[x^(2 / (e - 1)) a + 1]} /. G[[1]]]
Off[ParametricPlot::"ppcom"]
Off[FindRoot::"frnum"]

Show[ParametricPlot[FL[Exp[a], 10, 0.0001, 5, 7 / 3 + 0.3], {a, Log[0.001], Log[200000]},
   PlotRange → All, AxesLabel → None, DisplayFunction -> Identity],
  DisplayFunction -> $DisplayFunction, Ticks → None];
```

Figure 6 : Valeur asymptotique de λ

```
PR[d_, q_, z0_, T_] := Module[{M = 8 + 0.5 * Log[z0], α = d - 2 - 4/(q - 1),
   β = 2/(q - 1) (d - 2 - 2/(q - 1))},
  ParametricPlot[Evaluate[{Log[z[t]], x[t]} /. NDSolve[{x'[t] == y[t],
       x[0] == β^(1/(q-1)), y'[t] == -α y[t] + (β - z[t]) x[t] - Abs[x[t]]^(q-1) x[t], y[0] == 0,
       z'[t] == 2 z[t], z[0] == z0}, {x, y, z}, {t, 0, M}]], {t, 0, M}, Ticks → T]]
PR[5, 7 / 3 + 0.3, 0.01, None];
```



```
R[d_, q_, z0_] := Module[{M = 8 + 0.5 * Log[z0], α = d - 2 - 4/(q - 1),
    β = 2/(q - 1) (d - 2 - 2/(q - 1))}, FindRoot[Evaluate[x[s] /. NDSolve[
       {x'[t] == y[t], x[0] == β^(1/(q-1)), y'[t] == -α y[t] + (β - z[t]) x[t] - Abs[x[t]]^(q-1) x[t],
        y[0] == 0, z′[t] == 2 z[t], z[0] == z0}, {x, y, z}, {t, 0, M}]], {s, 3.5}]]
R[
 5,
 7 /
   3 +
  0.3,
 0.01]
```

```
{s → 3.5478}
```

```
Crit[d_, q_, z0_] := Module[{M = 8 + 0.5 * Log[z0], α = d - 2 - 4/(q - 1),
    β = 2/(q - 1) (d - 2 - 2/(q - 1))}, z0 * Exp[2 s /. R[d, q, z0]]]
Crit[5, 7 / 3 + 0.3, 0.01]
G[d_, u_] := Module[{F = Evaluate[Crit[d, (d + 2)/(d - 2) + u, 0.01]]}, F]
Plot[G[5, u], {u, 0.01, 10}];
```

```
12.0665
```